

## スプラインとデジタルフィルターによるデータスムージング のための Basic プログラム

斉藤 慎一 横井 孝志\*

### BASIC Programs of Cubic Spline Smoothing and Digital Filtering

Shin-ichi SAITO and TAKASHI YOKOI \*

This paper presents two BASIC programs, which are cubic spline smoothing and digital filtering, for the smoothing and differentiation of human motion data in cinematography.

#### I 目 的

映画分析による位置データに含まれるノイズ(誤差)は、人間の身体運動の場合には、その動作の信号周波数が比較的低次であるので 適当な低域フィルターを用いることにより軽減できると言われている<sup>4)</sup>。これまでバイオメカニクスの分野での映画分析に用いられてきたデータスムージングには大別すると、(1)単一の多項式、(2)区分的多項式、(3)移動荷重平均法、(4)デジタルフィルターなどの手法がある<sup>9)</sup>。

ところで、最近のマイクロコンピュータの発達と普及からすると、従来大型計算機やミニコンで比較的利用の多いデータスムージング法についてマイコン用の BASIC 言語でプログラムを書き換えておくことが役立つのではないと思われる。ここでは、上述の(2)と(4)の代表的なスムージング法をとり上げ利用上の注意などを合わせて紹介する。

#### II プログラムの説明

Appendix A には 3 次スプラインスムージングのための、Appendix B には Butterworth のデジタ

ルフィルターのためのプログラムを示した。なお、それぞれのプログラムの作成の際には Reinsh<sup>3)</sup>と Winter<sup>4)</sup>の資料を参考にした。

使用方法に移る前にマイコンのシステムとして本体の他にデータの記憶装置(フロッピーディスク、カセットレコーダ)、プリンターおよびモニターテレビ等が利用できるものとする(ここでは Apple II plus システムを例にとって説明するが、マイコンによっては入出力の形式が異なるので、その例としてシャープ MZ 80-B の場合について入出力の部分のプログラムを Appendix C に示した)。

また、映画フィルムからの位置データは既に座標変換され、実長に換算されておりフロッピーディスク等にデータ数、撮影速度などと共に次の順に記録されているものとする。

- (1)データ名: A\$, …… 1組のデータの識別名
- (2)データ数: N, …… 分析コマ数
- (3)撮影速度: F S, …… 1秒間の撮影コマ数
- (4)データ 1  
データ 2  
⋮  
データ n

\* 筑波大学大学院(博士課程)体育科学研究科

(a) スプラインの場合 (Appendix A)

行番号20~90はフロッピーディスクからのデータ入力部分, 行番号200~680はプログラムの本体そして行番号1000~1080は出力部分である。

20行の文字変数 A\$ はデータ名であり, 30行の入力パラメータ DY はスムージングの程度を決めるものであるが, 通常は一定値を用いる。位置データに含まれる平均的な誤差があらかじめわかっている場合にはその値 (標準偏差) を用いる。わかっていない場合には著者らの経験では, データの最大値と最小値の差の  $\frac{1}{500}$  ( $\frac{1}{100} \sim \frac{1}{1000}$ ) 程度の値を予測される誤差として用いる。

1060行の出力パラメータでは, 配列変数 RD(I) は Raw データ, 配列変数 FD(I) はスムージング後のデータ, 配列変数 V(I) は速度, 配列変数 A(I) は加速度また配列変数 RE(I) は Raw データとスムージングした後のデータとの差 (残差) である。

(b) デジタルフィルタの場合 (Appendix B)

行番号20~70はフロッピーディスクからのデータの入力部分, 行番号220~260はスムージング後の位置データを微分し, 速度と加速度を求めるためのプログラム, 行番号1000~1090は出力に関する部分, 行番号5000~5150はデジタルフィルタのプログラムの本体でありサブルーチンとして用いる。

20行の入力パラメータ HZ はデジタルフィルタのカットオフ周波数を与えるものである。著者らの経験では, 映画撮影速度の  $\frac{1}{5} \sim \frac{1}{10}$  程度の値を用いると良いようである。

1070行の出力パラメータのうちの配列変数はスプラインの場合と同じ意味である。

III 使用例

Pezzack ら<sup>2)</sup>のデータを用いて計算した結果を図1には位置データ (スプラインの場合) を, 図2 (スプライン) および図3 (デジタルフィルタ) には加速度の値を示した。

この場合, データ数は142個, サンプルレートは約0.02秒 ( $\frac{1}{49.75}$ ) であり, スプラインの場合にはDYを0.004に, デジタルフィルタの場合にはHZを8.0にした。




Fig. 1




Fig. 2




Fig. 3

フロッピーディスクのドライブ開始よりプリンターの計算結果の打ち出し終了までの所要時間はスプラインの場合は7分25秒であり、一方デジタルフィルタの場合では3分15秒であった（計算時間はそれぞれ4分30秒と40秒であった）。

#### IV 留意点など

基本的には、スムージングを行う以前の撮影段階あるいは読み取り段階での誤差の混入をできるだけ防止すること、またあらかじめ関数形がわかっているあとはそのパラメータを決定するだけであるようなデータには、これらのプログラムあるいは

データスムージングを用いる必要はないということである。

実際の利用では、普通一回の試行で良い結果を得ることは困難であり、試行錯誤的に進めることになるだろう。

一般に、映画分析によって得られた位置データから速度・加速度を求める場合、特に加速度の値が悪くなると言われている<sup>5)</sup>。スプラインのプログラムでは、データの両端での加速度はかならずゼロになることに注意しておく必要がある。デジタルフィルタのプログラムでも加速度の値の利用には慎重を要する。

本研究の要旨は第32回日本体育学会（於、神戸）のバイオメカニクス専門分科会シンポジウムで発表した。また、経費の一部は昭和56年度文部省科学研究費、奨励研究（A）、の補助によった。

#### 参 考 文 献

- 1) Cavanagh, P. R. Recent advances in instruction and methodology of biomechanical studies. In: P. V. Komi (ed), Biomechanics V-B, 399-411, University Park Press, 1976.
- 2) Pezzack, J. C., R. W. Norman and D. A. Winter An assessment of derivative determining techniques used for motion analysis. J. of Biomechanics 10, 377-382, 1977.
- 3) Reinsh, C. H. Smoothing by spline functions. Numerische Mathematik 10, 177-183, 1967.
- 4) Winter, D. A. Biomechanics of human movement. John Wiley & Sons, 1979.
- 5) 渋川侃二. 数値微分：体育学実験・演習概説（体育科教育研究会編），p.136，大修館書店，1979.

Appendix A

```

10 REM ***** CUBIC-SPLINE DATA SMOOTHING *****
20 INPUT "DATA NAME=";A$
30 INPUT "DELTA Y=";DY
40 E$ = CHR$(4): PRINT E$;"OPEN" + A$: PRINT E$;"READ" + A$
50 INPUT N,T
60 M = N + 2
70 DIM RD(M),FD(M),V(M),A(M),D(M),R(M),R1(M),R2(M),U(M),W(M),RE(M)
80 FOR I = 0 TO N: INPUT RD(I): NEXT I
90 PRINT E$;"CLOSE" + A$
200 Z = T:N1 = 1:N2 = N + 1:S = N:Q = DY
205 FOR I = N TO 0 STEP -1:RD(I + 1) = RD(I): NEXT I
210 M1 = N1 + 1:M2 = N2 - 1:H = Z:F = (RD(M1) - RD(N1)) / H
220 FOR I = M1 TO M2
230 G = H:H = Z:E = F:F = (RD(I + 1) - RD(I)) / H:FD(I) = F - E:T = 2 * (G
    + H) / 3
240 T1 = H / 3:R2(I) = Q / G:R(I) = Q / H:R1(I) = - Q / G - Q / H
250 NEXT I
260 FOR I = M1 TO M2
270 V(I) = R(I) * R(I) + R1(I) * R1(I) + R2(I) * R2(I)
280 A(I) = R(I) * R1(I + 1) + R1(I) * R2(I + 1):D(I) = R(I) * R2(I + 2)
300 NEXT I
310 F2 = - S
320 FOR I = M1 TO M2
330 R1(I - 1) = F * R(I - 1):R2(I - 2) = G * R(I - 2)
340 R(I) = 1 / (P * V(I) + T - F * R1(I - 1) - G * R2(I - 2))
350 U(I) = FD(I) - R1(I - 1) * U(I - 1) - R2(I - 2) * U(I - 2)
360 F = P * A(I) + T1 - H * R1(I - 1):G = H:H = D(I) * P
370 NEXT I
380 FOR I = M2 TO M1 STEP -1
390 U(I) = R(I) * U(I) - R1(I) * U(I + 1) - R2(I) * U(I + 2)
400 NEXT I
410 E = 0:H = 0
420 FOR I = N1 TO M2
430 G = H:H = (U(I + 1) - U(I)) / Z:W(I) = (H - G) * Q * Q:E = E + W(I) *
    (H - G)
440 NEXT I
450 G = - H * Q * Q:W(N2) = G:E = E - G * H:G = F2:F2 = E * P * P
460 IF F2 > = S OR F2 < = G GOTO 600
480 F = 0:H = (W(M1) - W(N1)) / Z
490 FOR I = M1 TO M2
500 G = H:H = (W(I + 1) - W(I)) / Z:G = H - G - R1(I - 1) * R(I - 1) - R2(
    I - 2) * R(I - 2)
510 F = F + G * R(I) * G:R(I) = G
515 NEXT I
520 H = E - P * F
525 IF H < = 0 GOTO 600
530 P = P + (S - F2) / ((SQR(S / E) + P) * H)
540 GOTO 320
600 FOR I = N1 TO N2
610 FD(I) = RD(I) - P * W(I):A(I) = U(I)
620 NEXT I
630 FOR I = N1 TO M2
640 H = Z:D(I) = (A(I + 1) - A(I)) / (3 * H)
650 V(I) = (FD(I + 1) - FD(I)) / H - (H * D(I) + A(I)) * H
660 NEXT I
670 FOR I = 1 TO N2:RE(I) = RD(I) - FD(I):A(I) = 2 * A(I): NEXT I
680 FS = INT(100 / Z + .5) / 100
1000 PR# 4:WRITE = 49312 + 256 * 4
1010 CALL WRITE:" ", / :
1020 CALL WRITE:" CUBIC-SPLINE SMOOTHING DATA NAME=",A$;A10, / :
1030 CALL WRITE:"SAMPLE RATE = ",FS;F9.4," DELTA Y = ",DY;FB.5, / :
1040 CALL WRITE:"NO. RAW DATA FILTERED DATA VELOCITY ACCELERATION
    RES.", / :
1050 FOR I = 1 TO N2
1060 CALL WRITE:I;I5,2X,RD(I);F10.5,2X,FD(I);F10.5,2X,V(I);F10.5,2X,A(I);
    F10.5,2X,RE(I);F10.7, / :
1070 NEXT I
1080 PR# 0: END

```

Appendix B

```

10 REM ***** BUTTERWORTH DIGITAL FILTER *****
20 INPUT "DATA NAME=";A$: INPUT "CUTOFF HZ=";HZ
30 E$ = CHR$(4): PRINT E$;"OPEN" + A$: PRINT E$;"READ" + A$
40 INPUT N,T: DIM RD(N + 2),FD(N + 2),V(N + 2),A(N + 2),RE(N + 2),D(N + 2)
50 FS = 1 / T
60 FOR I = 0 TO N: INPUT RD(I): NEXT I
70 PRINT E$;"CLOSE" + A$
200 FOR I = 0 TO N:D(I) = RD(I): NEXT I
210 GOSUB 5000
220 FOR I = 1 TO N - 1:V(I) = (FD(I + 1) - FD(I - 1)) / 2 / T: NEXT I
230 V(0) = (- 3 * FD(1) + 4 * FD(2) - FD(3)) / 2 / T:V(N) = (FD(N - 2) -
4 * FD(N - 1) + 3 * FD(N)) / 2 / T
240 FOR I = 1 TO N - 1:A(I) = (V(I + 1) - V(I - 1)) / 2 / T: NEXT I
250 A(0) = (- 3 * V(1) + 4 * V(2) - V(3)) / 2 / T:A(N) = (V(N - 2) - 4 *
V(N - 1) + 3 * V(N)) / 2 / T
260 FOR I = 0 TO N:RE(I) = RD(I) - FD(I): NEXT I
1000 PR# 4:WRITE = 49312 + 256 * 4
1010 CALL WRITE:" ", / :
1020 PRINT : PRINT : PRINT : CALL WRITE:" BUTTER WORTH DIGITAL FILTER
":
1030 CALL WRITE:" DATA NAME = ",A$:A15, / :
1040 CALL WRITE:5X,"SAMPLE RATE = ",FS;F9.4," CUTOFF HZ = ",HZ;F9.4, /
:
1050 CALL WRITE:"NO. RAW DATA FILTERED DATA VELOCITY ACCELERATION
RES.", / :
1060 FOR I = 0 TO N
1070 CALL WRITE:I + 1;15,2X,RD(I);F10.5,2X,FD(I);F10.5,2X,V(I);F10.5,2X,A
(I);F10.5,2X,RE(I);F10.5, / :
1080 NEXT I
1090 PR# 0: END
5000 W = SQR ( SQR ( SQR (2) + 1)) * TAN (3.14159265 * HZ / FS)
5030 C = 1 + SQR (2) * W + W * W
5040 C0 = W * W / C:C1 = 2 * (1 - W * W) / C
5050 C2 = (W * SQR (2) - 1 - W * W) / C
5060 FD(0) = D(0):FD(1) = D(1)
5070 FOR I = 2 TO N
5080 FD(I) = C0 * (D(I) + 2 * D(I - 1) + D(I - 2)) + C1 * FD(I - 1) + C2 *
FD(I - 2)
5090 NEXT I
5100 D(N) = FD(N):D(N - 1) = FD(N - 1)
5110 FOR I = N - 2 TO 0 STEP - 1
5120 D(I) = C0 * (FD(I) + 2 * FD(I + 1) + FD(I + 2)) + C1 * D(I + 1) + C2 *
D(I + 2)
5130 NEXT I
5140 FOR I = 0 TO N:FD(I) = D(I): NEXT I
5150 RETURN

```

Appendix C : シャープMZ-80Bの場合の入出力の変更について※

(1) スプライン

```
40 ROPEN/T A$
50 INPUT/T N,T
80 FOR I=0 TO N:INPUT/T RD(I):NEXT I
90 CLOSE/T

690 FOR I=1 TO N2
700 FD(I)=INT(FD(I)*100000+.5)/100000
710 V(I)=INT(V(I)*100000+.5)/100000
720 A(I)=INT(A(I)*100000+.5)/100000
730 NEXT I

1000 PRINT/P:PRINT/P:PRINT/P
1010 PRINT/P " CUBIC-SPLINE DATA SMOOTHING DATA NAME=";A$
1020 PRINT/P TAB(5);"SAMPLE RATE=";FS;TAB(25);"DELTA Y=";DY
1030 PRINT/P "NO.";TAB(5);"RAW DATA";TAB(15);"FILTERED DATA";
1040 PRINT/P TAB(31);"VELOCITY";TAB(42);"ACCELERATION";TAB(56);
1050 PRINT/P "RES."
1060 FOR I=1 TO N2
1070 PRINT/P I;TAB(5);RD(I);TAB(15);FD(I);TAB(31);V(I);TAB(42);
A(I);TAB(53);RE(I)
1080 NEXT I:END
```

(2) デジタルフィルター

```
30 ROPEN/T A$
40 INPUT/T N,T:DIM RD(N+2),V(N+2),A(N+2),RE(N+2),D(N+2),FD(N+2)
60 FOR I=0 TO N:INPUT/T RD(I):NEXT I
70 CLOSE/T

270 FOR I=0 TO N
280 RD(I)=INT(RD(I)*100000+.5)/100000
290 FD(I)=INT(FD(I)*100000+.5)/100000
300 V(I)=INT(V(I)*100000+.5)/100000
310 A(I)=INT(A(I)*100000+.5)/100000
320 NEXT I

1000 PRINT/P:PRINT/P:PRINT/P
1010 PRINT/P " BUTTERWORTH DIGITALFILTER DATA NAME=";A$
1020 PRINT/P TAB(5);"SAMPLE RATE=";FS;TAB(25);"CUTOFF HZ=";HZ
1030 PRINT/P "NO.";TAB(5);"RAW DATA";TAB(15);"FILTERDE DATA";
1040 PRINT/P TAB(31);"VELOCITY";TAB(42);"ACCELERATION";TAB(56);
1050 PRINT/P "RES."
1060 FOR I=0 TO N
1070 PRINT/P I+1;TAB(5);RD(I);TAB(15);FD(I);TAB(31);V(I);
1080 PRINT/P TAB(42);A(I);TAB(53);RE(I):NEXT I
1090 END
```

※ 入出力部分のみの変更であり、各プログラムの行番号は、Apple II 用のプログラムの行番号に対応する。なお、(1)の690 - 730行および(2)の270 - 320行は計算盤のケタ数をそろえるために付け加えた。